

SmartSpiceRF

Mixer Simulation

Tutorial

Introduction

Information presented in this Tutorial is based on the assumption that the user is familiar with the following:

- Basics of the computer operating system and hardware employed
- Basic terminology of semiconductor process and device operation
- Circuit design, schematic capture, and simulation

The simulation flow described in this Tutorial uses the Silvaco Process Design Kit (PDK) based on a fictional process called SBCD (Silvaco Bipolar Cmos Dmos). SBCD process includes the following devices: two low-voltage Cmos, two high-voltage Dmos, four resistors, capacitor, and inductor. SBCD libraries provide all the data needed to demonstrate a real design flow.

This Tutorial presents a front-end design flow for a Mixer circuit and guides the user through the following circuit design steps:

- Schematic Capture with Gateway
- Simulation with SmartSpiceRF
- Results post-processing with SmartView

The following Silvaco EDA tools are needed to work with Tutorial:

- GATEWAY
- SMARTSPICERF
- SMARTVIEW

1: GATEWAY Schematic Capture

Start GATEWAY and load the workspace file `mixer.workspace` by selecting **File→Open→Workspace**. When the browser window appears, navigate to the directory where RF Demo PDK Examples were installed, descent into `./MIXER` directory, and select the `mixer.workspace` file.

1.1: Loading Mixer Schematic

The Gilbert cell mixer simulation will be setup with an RF input frequency of 990MHz, a 1 GHz Local Oscillator (LO) generating a 10 MHz Intermediate Frequency (IF) output.

1.1.1: To load the Mixer example circuit schematic, select **File->Open->Schematic**. When the file browser appears, navigate to the `./MIXER` directory and select the file `mixer.schlr`. The Mixer example circuit schematic will appear in the window (Figure 1).




Figure 1: Mixer Schematic

1.2: Generating Netlists

Netlists are the text files used to describe device connectivity and element properties of the circuit. GATEWAY can create two different netlist formats: SMARTSPICE for circuit simulation, or GUARDIAN for layout design. See the GATEWAY USER'S MANUAL for more information on these formats.

1.2.1: To generate netlist in SMARTSPICE format, either select **Simulation→Create Netlist**, or click on **SmartSpice Netlist** icon (Figure 2) in **Tool Bar**.



Figure 2: Create Netlist Icon

SMARTSPICE netlist then will appear in a new window:

```

File Edit View Bookmarks Spice Help
* Gateway 2.8.25.R Spice Netlist Generator
* Workspace name:      D:\susanw\silvaco\pdk\simucad-radiofrequency-demo\1.1.5.R\gateway\simucad-radiofreq
* Simulation name:      D:\susanw\silvaco\pdk\simucad-radiofrequency-demo\1.1.5.R\gateway\examples\mixer\mi
* Simulation timestamp: 30-Jul-2009 16:10:54

* Schematic name: mixer
I1 VDD NET6 DC 60u
M1 NET8 NET6 GND GND mn W=10U L=1u AS=6.6P AD=6.6P PS=21.32U PD=21.32U M=20
M2 NET6 NET6 GND GND mn W=10U L=1u AS=6.6P AD=6.6P PS=21.32U PD=21.32U M=1
V1 VDD GND DC 3
V4 NET7 GND DC 1.5
V6 NET11 GND DC 1
XC1 LOIN_M NET3 mimcap LT=6.5728U M=1
XC2 IFOUT_M IFOUT_P mimcap LT=3.2864U M=1
XC3 LOIN_P NET1 mimcap LT=6.5728U M=1
XC4 IFOUT_P IFOUT_M mimcap LT=3.2864U M=1
XM1 IFOUT_P NET1 NET5 NET5 rfnmos2v LR=240N WR=5U NR=8 M=1
XM4 IFOUT_P NET3 NET4 NET4 rfnmos2v LR=240N WR=5U NR=8 M=1
XM5 IFOUT_M NET1 NET4 NET4 rfnmos2v LR=240N WR=5U NR=8 M=1
XM6 IFOUT_M NET3 NET5 NET5 rfnmos2v LR=240N WR=5U NR=8 M=1
XM7 NET4 RFIN_P NET8 NET8 rfnmos2v LR=240N WR=5U NR=16 M=1
XM8 NET5 RFIN_M NET8 NET8 rfnmos2v LR=240N WR=5U NR=16 M=1
XR1 NET7 NET3 rpoly W=2U L=131.34U nc=3 M=1
XR2 VDD IFOUT_P rpoly W=2U L=25.207U nc=3 M=1
XR3 VDD IFOUT_M rpoly W=2U L=25.207U nc=3 M=1
XR4 NET11 RFIN_P rpoly W=2U L=131.34U nc=3 M=1
XR5 NET11 RFIN_M rpoly W=2U L=131.34U nc=3 M=1
XR6 NET7 NET1 rpoly W=2U L=131.34U nc=3 M=1

* Global Nodes Declarations
.GLOBAL GND VDD

* End of the netlist
  
```

Figure 3: Mixer Netlist

1.3: Simulator Setup

Check in GATEWAY that your preferred simulator is set to SMARTSPICERF. In GATEWAY click on **Edit -> Preferences** to open the preferences setup window (Figure 4). Choose **Tools-> Simulator** and set (check) Simulator to SMARTSPICERF. Version number is set to Default (the latest installed version) in **Version** window. You can define any specific version too.




Figure 4: Simulator Preferences Window

1.4: Control File

SMARTSPICERF needs more than just a circuit netlist to perform a meaningful simulation. It needs Circuit Netlist, Voltage or Current stimulus, Options, Analysis Statements, and active device Model parameters or Libraries. All this information comes together in the form of an Input Deck file (*.in, *.inp, *.cir, *.sp, *.sm, *.scs, etc.), which provides SMARTSPICERF with all needed information to run simulation and generate output data.

1.4.1: Click on the simulation **Tool Bar** icon **Edit Control File** (Figure 5):



Figure 5: Edit Control File Icon

The control file will be opened in the text editor Sedit window. The control file mixer.ctr (Figure 6) consists of path to SILVACO RF Demo PDK model library file as well as SMARTSPICERF .OP and .SPECTRAL Analyses statements. The control file combined with the circuit netlist and list of vectors to be saved creates an Input Deck for SMARTSPICERF.

```

File Edit View Bookmarks Spice Help
[Icons] [AC] [DC] [IN] [H] [N] [OP] [PZ] [X] [Y] [F]
*
.inc ../../models/sbcd_rf.lib
.OP
*
.PARAM prf=-40_dBm v1o = 0.3
.OPTIONS EXPERT=777

*#com
*****
* Output Harmonics
*****
Pif IFOUT_P IFOUT_M R=10K
Prf RFIN_P RFIN_M MTS1(prf 0 1 -1 0)
V1op LOIN_P GND mag1=v1o
V1om LOIN_M GND mag1=v1o phase1=180
*----- Pdisto -----
.SPECTRAL
+ fund1=1Ghz nharm1=5
+ fund2=10MegHz nharm2=15
+ solver=mfgmres
+ lrs_tol=0.5
+ oversample=2
+ annotate=3 waves
*
.PROBE SPECT_SP vm(IFOUT_P, IFOUT_M)
*
*#endcom

#com
*****
* Noise Figure
*****
Pif IFOUT_P IFOUT_M R=10K
Prf RFIN_P RFIN_M
V1op LOIN_P GND mag1=v1o
V1om LOIN_M GND mag1=v1o phase1=180
*----- Noise Figure Measurement -----
.HNOISE Pif Prf
+ DEC 10 10MHz 100MHz
+ fund=10Hz
+ refSB=-1
+ Kmex=10
+ Annotate=3
+ WAVES
*
*#endcom
  
```

Figure 6: Mixer Circuit Control File

2: SmartSpiceRF Simulation

This Tutorial illustrates how SmartSpiceRF can simulate mixers and provide basic measurements.

2.1: Quasi Periodic Steady-State Analysis

The input sources for this analysis are defined as Port and Voltage sources. Using a Port source allows us to input or measure rf signals in terms of power (dBm) while specifying an associated port resistance. These are helpful for power matching, gain and noise measurements, shown later.

The Mixer circuit control file has the following stimulus:

```
*****
Pif IFOUT_P IFOUT_M R=10K
Prf RFIN_P RFIN_M MTS1(prf 0 1 -1 0)
Vlop LOIN_P GND mag1f=vlo
Vlom LOIN_M GND mag1f=vlo phase1f=180
*****
```

Pif is a resistive port load across the IF output with load 10K ohms.

Prf is the RF input source defined by the coefficients in it's Multi-Tone Spectral (MTS) source for the fundamental tones defined in a Spectral analysis statement. The signal power is prf (dBm), phase is 0, and the frequency will be $(1)*Fund1+(-1)*Fund2+(0)*Fund3$.

2.1.1: Uncomment the .SPECTRAL analysis statement to make the analysis active.

```
*****
* Output Harmonics
*----- Pdisto -----
.SPECTRAL
+ fund1=1Ghz nharm1=5
+ fund2=10MegHz nharm2=15
+ solver=mfgmres
+ lrs_tol=0.5
+ oversample=2
+ annotate=3 waves
*
.PROBE SPECT_SP vm(IFOUT_P, IFOUT_M)
*****
```

In the .SPECTRAL analysis we define the fundamental signals in the circuit. There can be up to 3 fundamentals defined, here we have only defined 2 so the third one is 0Hz. 1GHz for the LO and 10MHz for the IF frequency. Prf input frequency is therefore $1\text{GHz} - 10\text{MHz} + 0\text{Hz} = 990\text{MHz}$ when we apply the Prf MTS coefficients to the .SPECTRAL analysis fundamentals.

Solver and lrs_tol are analysis parameters for the solver. We are loosening the large signal convergence tolerance by specifying lrs_tol=0.5 because the default value is 1e-3. In this particular circuit simulation we could also use the default value as the circuit converges easily.

Oversample multiplies the number of points used in the time-domain waveforms to represent the non-linear devices while reducing the FFT aliasing and improving simulation accuracy and robustness.

Annotate=3 generates the necessary spectral data and time-domain waveforms from the simulation run.

2.1.2: Click on  (**Gateway -> Simulation -> Run**) to begin the SMARTSPICERF simulation.

2.2: QPSS Simulation Results

SmartSpiceRF outputs simulation results of Large-Signal QPSS analysis and statistic information into Output window and creates number of plots with Waveforms, Spectra, Measurement results, etc. Simulation results in form of plots will be loaded into SMARTVIEW, and SMARTVIEW **Data Browser** window will be open.

2.2.1: Open plot `spect_sp1`, then select vector `vm(ifout_p, ifout_m)` and click **Plot**. Place cursor on the Chart **and Click** on Right mouse button to display **Cartesian Properties**. Select **dBm(MAG)** on **Data Map Type** window. From the menu **Chart**, select **Bars** and unselect **Lines**.

The output harmonics simulation plot is shown in Figure 7.




Figure 7: Mixer Output Spectra Plot

2.3: Noise Figure Simulation

Periodic Steady-State Noise Analysis (.HNOISE Analysis) is a small-signal analysis that must follow a Steady-State analysis. This analysis is used to compute mixer noise figure.

The input sources for this analysis are defined as Port and Voltage sources:

```
*****
Pif IFOUT_P IFOUT_M R=10K
Prf RFIN_P RFIN_M
Vlop LOIN_P GND mag1=vlo
Vlom LOIN_M GND mag1=vlo phase1=180
*****
```

For the .HNoise analysis we do not need to define an active RF input signal because a small-signal ac source, Prf here, is sufficient.

2.3.1: Uncomment .HNOISE analysis statement to calculate Noise Figure:

```
*****
* Noise Figure
*----- Noise Figure Measurement -----
.HNOISE Pif Prf
+ DEC 10 10MHz 100MHz
+ fund=1GHz
+ refSB=-1
+ Kmax=10
+ Annotate=3
+ WAVES
*****
```

The HNOISE analysis output (Pif) and input reference(Prf) ports are defined. A Periodic Steady State (PSS) analysis is automatically run by HNOISE using 1GHz as the fundamental frequency and a default of 4 harmonics are used to calculate the Periodic Operating Point of the circuit. After the PSS analysis has converged an output frequency sweep from 10MHz to 100MHz is specified with the circuit fundamental being 1GHz. The reference sideband for input referred noise calculations is -1 which equates to an input frequency band of 10M->100MHz - 1GHz = -990MHz -> -900MHz or 900M->990MHz at the RF input port Prf. The maximum number of noise sidebands (KMAX) is defined as 10 so every sideband from -10....-1, 0, +1....+10 will have it's noise contributions included.

2.3.2: Click on  (**Gateway -> Simulation -> Run**) to begin the SMARTSPICERF simulation.

2.4: Noise Figure Simulation Results

Simulation results in form of plots will be loaded into SMARTVIEW, and SMARTVIEW **Data Browser** window will be open.

2.4.1: Open plot hnoise1, then select vector “NFssb” and click **Plot**. The noise figure plot is shown in Figure 8.




Figure 8: Noise Figure Plot

2.5: S-parameters Simulation

Periodic Steady-State NET Analysis (.HNET Analysis) is a small-signal analysis, which is used to compute scattering parameters (S-parameters) for two-port circuits that exhibit frequency translation.

The input sources for this analysis are defined as Port and Voltage sources:


```
*****
Pif IFOUT_P IFOUT_M R=10K
Prf RFIN_P RFIN_M
Vlop LOIN_P GND mag1=vlo
Vlom LOIN_M GND mag1=vlo phase1=180
*****
```

Again the rf input port Prf only needs to be a small-signal source so we have not specified a Multi-Tone Signal.

2.5.1: Uncomment the .HNET analysis statement.

```
*****
* S-Parameter
*----- S-parameter Measurement -----
.HNET Prf Pif
+ start=-200MegHz stop=200MegHz step=10MegHz
+ FTYPE=REL
+ FUND=1GHz nharm=5
+ PORTHARM=(1,0)
+ FAXIS=OUT
+ annotate=3
+ waves
*****
```

HNET will make a frequency sweep from -200MHz to +200MHz relative offset from the 1GHz fundamental (800MHz - 1200MHz) on the input port 1 to -200MHz to +200MHz on the output to calculate S-parameters .

2.5.2: Click on  (Gateway -> Simulation -> Run) to begin the SMARTSPICERF simulation.

2.6: S-parameters Simulation Results

Simulation results in form of plots will be loaded into SMARTVIEW, and SMARTVIEW **Data Browser** window will be open.

2.6.1: Select SMARTVIEW **Chart**→**Create**→**Extended Smith** or click on the **Extended Smith Chart** icon (Figure 9).



Figure 9: Extended Smith Chart Icon

2.6.2: Open Plot `hnet1`, then select vector “`s22_h1h0`”, and click **Plot**. Figure 10 shows `s22` plot on Smith Chart.




Figure 10: S22 Smith Chart Plot

2.7: Conversion Gain Simulation

Periodic Steady-State Transfer Function Analysis (.HTF Analysis) is a small-signal analysis that follows a Periodic Steady-State analysis. It is used to compute mixer conversion gain.

The input sources for this analysis are defined as Port and Voltage sources:


```
*****
Pif IFOUT_P IFOUT_M R=10K
Prf RFIN_P RFIN_M
Vlop LOIN_P GND mag1=vlo
Vlom LOIN_M GND mag1=vlo phase1=180
*****
```

HTF is also a small-signal analysis calculated about the large signal periodic operating point of it's initial PSS analysis, so the RF input only needs be a small-signal source during the frequency translating transfer function analysis.

2.7.1: Uncomment .HTF analysis statement for Conversion Gain calculation:

```
*****
* Conversion Gain
*----- TF analysis -----
.HTF v(IFOUT_P, IFOUT_M)
+ LIN 51 1Meg 301Meg
+ FUND=1GHz
+ SB=(-1, 1)
+ annotate=3
+ waves
*
.let HTF Prf_Gain1h=db(tfh1_prf)
.let HTF Prf_Gain_1h=db(tfh_1_prf)
*****
```

HFT will run a PSS analysis using 1GHz as it's fundamental. After the Periodic Operating Point is calculated a sweep of the output frequency from 1MHz to 301MHz in 51 linear steps is run and the transfer functions from circuit sources to the output nodes IFOUT_P, IFOUT_M is calculated. For each source the frequency translations from the sidebands -1 (1MHz->301MHz -1GHz = 699MHz->999MHz) and sideband +1 (1MHz->301MHz + 1GHz = 1001M->1301MHz) are calculated. A .LET statement is used to make a db20 calculation of the voltage transfer functions for easy plotting later.

2.7.2: Click on  (**Gateway -> Simulation -> Run**) to begin the SMARTSPICERF simulation.

2.8: Conversion Gain Simulation Results

Simulation results in form of plots will be loaded into SMARTVIEW, and SMARTVIEW **Data Browser** window will be open.

2.8.1: Open plot htf1, then select vectors “prf_gain1h” and “prf_gain_1h”, and click **Plot**. The mixer conversion gain plot is shown in Figure 11.




Figure 11: Mixer Conversion Gain Plot

2.9: 1dB Compression Point Calculation

The Input Referred 1dB Compression Point is that input signal level at which the actual gain departs from the theoretical gain by -1dB. SMARTSPICERF allows you to compute Input Referred 1dB Compression Point using the measurement statement COMPR1DB.

The input sources for this analysis are defined as Port and Voltage sources:


```
Pif IFOUT_P IFOUT_M R=10K
Prf RFIN_P RFIN_M MTS1(prf 0 1 -1 0)
Vlop LOIN_P GND mag1f=vlo
Vlom LOIN_M GND mag1f=vlo phase1f=180
*****
```

For the SPECTRAL analysis we will need to apply a large signal RF input to measure the effect of the input signal compressing the gain of the Mixer. The MTS definition means a 1GHz - 10MHz+0Hz =990MHz input signal is generated at the RF input by Prf port.

2.9.1: Uncomment the .SPECTRAL analysis statement to calculate the 1dB compression point.

```
* 1db Compression Point
*----- Calculation the 1 dB Compression Point -----
.SPECTRAL
+ fund1=1Ghz nharm1=5
+ fund2=10MegHz nharm2=10
+ solver=mfgmres
+ lrs_tol=0.5
+ oversample=2
+ annotate=3
+ waves
+ SWEEP prf -40 0 5
*
*.PROBE SPECT_SP vm(IFOUT_P, IFOUT_M)
*
.MEASURE SPECT_SP dbOUT FIND vdb(IFOUT_P, IFOUT_M) AT=10Meg
.MEASURE MEAS_r_1dbCompression COMPR1DB dbOUT EP=-35
*****
```

SPECTRAL analysis will run a QPSS analysis for each step of the Prf input power sweep from -40 (dBm) to 0 (dBm) and the measurements are taken of the output power at 10MHz with the FIND function. These measurements are then used to calculate the 1dB compression point using the SMARTSPICERF function COMPR1DB. The scaled result is saved in r_1dbCompression.

2.9.2: Click on  (Gateway -> Simulation -> Run) to begin the SMARTSPICERF simulation.

2.10: 1dB Compression Point Simulation Results

Simulation results in form of plots will be loaded into SMARTVIEW, and SMARTVIEW **Data Browser** window will be open.

2.10.1: Open plot meas1, then select vector “dbout”, and click **Plot**.

The 1dB compression point plot is shown in Figure 12.




Figure 12: 1dB Compression Point Plot

2.10.2: In SMARTSPICERF, you can see the actual values that are being plotted. To view these, select vector “r_1dbcompression” from “meas1” and press **View Data** and 1dB compression point value is shown in Figure 13.




Figure 13: View Data Window: 1 dB Compression Point = -12.09 dBm

2.11: Third-Order Intercept Point Calculation

The input sources for this analysis are defined as Port and Voltage sources:

```
*****
```

```
Pif IFOUT_P IFOUT_M R=10K
Prf RFIN_P RFIN_M MTS1(prf 0 1 -1 0) MTS2(prf 0 1 0 -1)
Vlop LOIN_P GND mag1f=vlo
Vlom LOIN_M GND mag1f=vlo phase1f=180
*****
```


For Third-Order Intercept we need to add a second input signal MTS2 at 1GHz + 0Hz - 10.1MHz = 989.9MHz with the same power level as the 990MHz signal. A third fundamental is added in the SPECTRAL analysis definition, fund3=10.1MHz.

2.11.1: Uncomment .SPECTRAL analysis statement from the section IIP3 to calculate the Third-Order Intercept Point (IIP3).

```
*****
```

```
* IIP3
*----- Calculation the IIP3 -----
.SPECTRAL
+ fund1=1Ghz nharm1=5
+ fund2=10MegHz nharm2=2
+ fund3=10.1MegHz nharm3=2
+ solver=mfgmres
+ lrs_tol=0.5
+ oversample=2
+ annotate=3
+ waves
+ SWEEP prf -40 0 5
*
.MEASURE SPECT_SP POUT_1 FIND vdb(IFOUT_P, IFOUT_M) AT=10Meg
.MEASURE SPECT_SP POUT_3 FIND vdb(IFOUT_P, IFOUT_M) AT=10.2Meg
.MEASURE MEAS_resIP3 IP3 POUT_1 POUT_3 EP=-35 GAIN=2
*****
```

The function SMARTSPICERF function IP3 is used to calculate the Input referred third-order intercept point.

2.11.2: Click on  (Gateway -> Simulation -> Run) to begin the SMARTSPICERF simulation.

2.12: Third-Order Intercept Point Simulation Results

Simulation results in form of plots will be loaded into SMARTVIEW, and SMARTVIEW **Data Browser** window will be open.

2.12.1: Open plot meas1, then select both vectors “pout_1” and “pout_3”, and click **Plot**. The IP3 simulation plot is shown in Figure 14.




Figure 14: IP3 Simulation Plot

2.12.2: In SMARTSPICERF, you can see the actual values that are being plotted. To view its, select vector “resip3” from “ip3_meas1” and press **View Data**. IP3 value will be shown in window as on Figure 15.




Figure 15: View Data Window: IIP3 = -2.33 dBm